

Challenge November – 2014

Who killed Agatha ?

A solution with OPL CPLEX

by Alex Fleischer afleischer@fr.ibm.com

OPL (Optimization Programming Language) is an abstract modeling language that helps model easily optimization problems that can be solved both with IBM CPLEX linear programming and IBM CPLEX constraint programming CPOptimizer (CPO)

Someone in Dreadsbury Mansion killed Aunt Agatha. Agatha, the butler, and Charles live in Dreadsbury Mansion, and are the only ones to live there. A killer always hates, and is no richer than his victim. Charles hates noone that Agatha hates. Agatha hates everybody except the butler. The butler hates everyone not richer than Aunt Agatha. The butler hates everyone whom Agatha hates. Noone hates everyone. Who killed Agatha?

In OPL CPLEX with Linear Programming

```
//Someone in Dreadsbury Mansion killed Aunt Agatha. Agatha,
//the butler, and Charles live in Dreadsbury Mansion, and are the only ones to
//live there.

{string} people={"Agatha","Butler","Charles"};

dvar boolean hates[people][people];
dvar boolean isRicher[people][people];

dvar boolean isTheKiller[people];

subject to
{
    // someone killed Agatha
    sum(i in people) isTheKiller[i]==1;

    // isRicher is an order
    forall(i in people) isRicher[i][i]==0;
    forall(ordered i,j in people) isRicher[i][j]+isRicher[j][i]==1;
    forall(i,j,k in people) (isRicher[i][j]==1 && isRicher[j][k]==1) =>
(isRicher[i][k]==1);

    //A killer always hates, and is no richer than his victim.
    forall(i in people) (isTheKiller[i]==1) => (hates[i]["Agatha"]==1);
    forall(i in people) (isTheKiller[i]==1) => (isRicher[i]["Agatha"]==0);
```

```

// Charles hates noone that Agatha hates.
forall(i in people) (hates["Agatha"][i]==1) => (hates["Charles"][i]==0);

//Agatha hates everybody except the butler.
forall(i in people :i!="Butler") hates["Agatha"][i]==1;

//The butler hates everyone not richer than Aunt Agatha.
forall(i in people) (isRicher[i]["Agatha"]==0) => (hates["Butler"][i]==1);

//The butler hates everyone whom Agatha hates.
forall(i in people) (hates["Agatha"][i]==1) => (hates["Butler"][i]==1);

//Noone hates everyone.
forall(i in people) sum(j in people) hates[i][j]<=card(people)-1;

}

execute
{
  for(var i in people) if (isTheKiller[i]==1) writeln("The killer is ",i);
}

```

Gives

The killer is Agatha

And to prove there is no other possible solution we can add

```
isTheKiller["Agatha"]==0;
```

And then we do not get any solution.

NB:

To use constraint programming instead of linear programming just add

```
using CP;
```

At the beginning of the model.